(** * Lists: Working with Structured Data *)

Require Export Induction.
Module NatList.

(* ################################################################# *)
(** * Pairs of Numbers *)

(** In an [Inductive] type definition, each constructor can take
    any number of arguments -- none (as with [true] and [O]), one (as
    with [S]), or more than one, as here: *)

Inductive natprod : Type :=
| pair : nat -> nat -> natprod.

(** This declaration can be read: "There is just one way to
    construct a pair of numbers: by applying the constructor [pair] to
    two arguments of type [nat]." *)

Check (pair 3 5).

(** Here are two simple functions for extracting the first and
    second components of a pair.  The definitions also illustrate how
    to do pattern matching on two-argument constructors. *)

Definition fst (p : natprod) : nat :=
  match p with
  | pair x y => x
  end.

Definition snd (p : natprod) : nat :=
  match p with
  | pair x y => y
  end.

Compute (fst (pair 3 5)).
(* ===> 3 *)

(** Since pairs are used quite a bit, it is nice to be able to
    write them with the standard mathematical notation [(x,y)] instead
    of [pair x y].  We can tell Coq to allow this with a [Notation]
    declaration. *)

Notation "( x , y )" := (pair x y).

(** The new pair notation can be used both in expressions and in
    pattern matches (indeed, we've actually seen this already in the
    [Basics] chapter, in the definition of the [minus] function --
    this works because the pair notation is also provided as part of
    the standard library): *)

Compute (fst (3,5)).

Definition fst' (p : natprod) : nat :=
  match p with
  | (x,y) => x
  end.

Definition snd' (p : natprod) : nat :=
  match p with
  | (x,y) => y
  end.

Definition swap_pair (p : natprod) : natprod :=
  match p with
  | (x,y) => (y,x)
  end.

(** Let's try to prove a few simple facts about pairs.

    If we state things in a particular (and slightly peculiar) way, we
    can complete proofs with just reflexivity (and its built-in
    simplification): *)

Theorem surjective_pairing' : forall (n m : nat),
  (n,m) = (fst (n,m), snd (n,m)).
Proof.
  reflexivity.  Qed.

(** But [reflexivity] is not enough if we state the lemma in a more
    natural way: *)

Theorem surjective_pairing_stuck : forall (p : natprod),
  p = (fst p, snd p).
Proof.
  simpl. (* Doesn't reduce anything! *)
Abort.

(** We have to expose the structure of [p] so that [simpl] can
    perform the pattern match in [fst] and [snd].  We can do this with
    [destruct]. *)

Theorem surjective_pairing : forall (p : natprod),
  p = (fst p, snd p).
Proof.
  intros p.  destruct p as [n m].  simpl.  reflexivity.  Qed.

(** Notice that, unlike its behavior with [nat]s, [destruct]
    generates just one subgoal here.  That's because [natprod]s can
    only be constructed in one way. *)

(** **** Exercise: 1 star (snd_fst_is_swap)  *)
Theorem snd_fst_is_swap : forall (p : natprod),
  (snd p, fst p) = swap_pair p.
Proof.
intros.
destruct p.
simpl.
reflexivity.
Qed.

(** **** Exercise: 1 star, optional (fst_swap_is_snd)  *)
Theorem fst_swap_is_snd : forall (p : natprod),
  fst (swap_pair p) = snd p.
Proof.
intros. destruct p.
simpl. reflexivity.
Qed.
(* ################################################################# *)
(** * Lists of Numbers *)

(** Generalizing the definition of pairs, we can describe the
    type of _lists_ of numbers like this: "A list is either the empty
    list or else a pair of a number and another list." *)

Inductive natlist : Type :=
  | nil  : natlist
  | cons : nat -> natlist -> natlist.

(** For example, here is a three-element list: *)

Definition mylist := cons 1 (cons 2 (cons 3 nil)).

(** As with pairs, it is more convenient to write lists in
    familiar programming notation.  The following declarations
    allow us to use [::] as an infix [cons] operator and square
    brackets as an "outfix" notation for constructing lists. *)

Notation "x :: l" := (cons x l)
                     (at level 60, right associativity).
Notation "[ ]" := nil.
Notation "[ x ; .. ; y ]" := (cons x .. (cons y nil) ..).

(** It is not necessary to understand the details of these
    declarations, but in case you are interested, here is roughly
    what's going on.  The [right associativity] annotation tells Coq
    how to parenthesize expressions involving several uses of [::] so
    that, for example, the next three declarations mean exactly the
    same thing: *)

Definition mylist1 := 1 :: (2 :: (3 :: nil)).
Definition mylist2 := 1 :: 2 :: 3 :: nil.
Definition mylist3 := [1;2;3].

(** The [at level 60] part tells Coq how to parenthesize
    expressions that involve both [::] and some other infix operator.
    For example, since we defined [+] as infix notation for the [plus]
    function at level 50,

  Notation "x + y" := (plus x y)
                      (at level 50, left associativity).

    the [+] operator will bind tighter than [::], so [1 + 2 :: [3]]
    will be parsed, as we'd expect, as [(1 + 2) :: [3]] rather than [1
    + (2 :: [3])].

    (Expressions like "[1 + 2 :: [3]]" can be a little confusing when
    you read them in a [.v] file.  The inner brackets, around 3, indicate
    a list, but the outer brackets, which are invisible in the HTML
    rendering, are there to instruct the "coqdoc" tool that the bracketed
    part should be displayed as Coq code rather than running text.)

    The second and third [Notation] declarations above introduce the
    standard square-bracket notation for lists; the right-hand side of
    the third one illustrates Coq's syntax for declaring n-ary
    notations and translating them to nested sequences of binary
    constructors. *)

(* ----------------------------------------------------------------- *)
(** *** Repeat *)

(** A number of functions are useful for manipulating lists.
    For example, the [repeat] function takes a number [n] and a
    [count] and returns a list of length [count] where every element
    is [n]. *)

Fixpoint repeat (n count : nat) : natlist :=
  match count with
  | O => nil
  | S count' => n :: (repeat n count')
  end.

(* ----------------------------------------------------------------- *)
(** *** Length *)

(** The [length] function calculates the length of a list. *)

Fixpoint length (l:natlist) : nat :=
  match l with
  | nil => O
  | h :: t => S (length t)
  end.

(* ----------------------------------------------------------------- *)
(** *** Append *)

(** The [app] function concatenates (appends) two lists. *)

Fixpoint app (l1 l2 : natlist) : natlist :=
  match l1 with
  | nil    => l2
  | h :: t => h :: (app t l2)
  end.

(** Actually, [app] will be used a lot in some parts of what
    follows, so it is convenient to have an infix operator for it. *)

Notation "x ++ y" := (app x y)
                     (right associativity, at level 60).

Example test_app1:             [1;2;3] ++ [4;5] = [1;2;3;4;5].
Proof. reflexivity.  Qed.
Example test_app2:             nil ++ [4;5] = [4;5].
Proof. reflexivity.  Qed.
Example test_app3:             [1;2;3] ++ nil = [1;2;3].
Proof. reflexivity.  Qed.

(* ----------------------------------------------------------------- *)
(** *** Head (with default) and Tail *)

(** Here are two smaller examples of programming with lists.
    The [hd] function returns the first element (the "head") of the
    list, while [tl] returns everything but the first
    element (the "tail").
    Of course, the empty list has no first element, so we
    must pass a default value to be returned in that case.  *)

Definition hd (default:nat) (l:natlist) : nat :=
  match l with
  | nil => default
  | h :: t => h
  end.

Definition tl (l:natlist) : natlist :=
  match l with
  | nil => nil
  | h :: t => t
  end.

Example test_hd1:             hd 0 [1;2;3] = 1.
Proof. reflexivity.  Qed.
Example test_hd2:             hd 0 [] = 0.
Proof. reflexivity.  Qed.
Example test_tl:              tl [1;2;3] = [2;3].
Proof. reflexivity.  Qed.

(* ----------------------------------------------------------------- *)
(** *** Exercises *)

(** **** Exercise: 2 stars, recommended (list_funs)  *)
(** Complete the definitions of [nonzeros], [oddmembers] and
    [countoddmembers] below. Have a look at the tests to understand
    what these functions should do. *)

Fixpoint nonzeros (l:natlist) : natlist:=
  match l with
  | [] => []
  | O::xs => nonzeros xs
  | S(n)::xs => S(n)::(nonzeros xs)
  end.

Example test_nonzeros:
  nonzeros [0;1;0;2;3;0;0] = [1;2;3].
Proof. reflexivity. Qed.

Fixpoint oddmembers (l:natlist) : natlist:=
  match l with
  | [] => []
  | x::xs => match oddb x with
             | true => x::oddmembers xs
             | false => oddmembers xs
             end
  end.

Example test_oddmembers:
  oddmembers [0;1;0;2;3;0;0] = [1;3].
(* FILL IN HERE *)
Proof.
  reflexivity.
Qed.
  (* GRADE_THEOREM 0.5: NatList.test_oddmembers *)

Definition countoddmembers (l:natlist) : nat:= length (oddmembers l).

Example test_countoddmembers1:
  countoddmembers [1;0;3;1;4;5] = 4.
Proof. reflexivity. Qed.
Example test_countoddmembers2:
  countoddmembers [0;2;4] = 0.
Proof. reflexivity. Qed.

Example test_countoddmembers3:
  countoddmembers nil = 0.
Proof. reflexivity. Qed.

(** **** Exercise: 3 stars, advanced (alternate)  *)
(** Complete the definition of [alternate], which "zips up" two lists
    into one, alternating between elements taken from the first list
    and elements from the second.  See the tests below for more
    specific examples.

    Note: one natural and elegant way of writing [alternate] will fail
    to satisfy Coq's requirement that all [Fixpoint] definitions be
    "obviously terminating."  If you find yourself in this rut, look
    for a slightly more verbose solution that considers elements of
    both lists at the same time.  (One possible solution requires
    defining a new kind of pairs, but this is not the only way.)  *)

Fixpoint alternate (l1 l2 : natlist) : natlist:=
  match l1 with
  | [] => match l2 with
            | [] => []
            | y::ys => l2
            end
  | x::xs => match l2 with
             | []=> l1
             | y::ys => x::y::alternate xs ys
             end
  end.

Example test_alternate1:
  alternate [1;2;3] [4;5;6] = [1;4;2;5;3;6].
(* FILL IN HERE *)
Proof. reflexivity. Qed.

Example test_alternate2:
  alternate [1] [4;5;6] = [1;4;5;6].
Proof. reflexivity. Qed.

Example test_alternate3:
  alternate [1;2;3] [4] = [1;4;2;3].
Proof. reflexivity. Qed.

Example test_alternate4:
  alternate [] [20;30] = [20;30].
Proof. reflexivity. Qed.

(* ----------------------------------------------------------------- *)
(** *** Bags via Lists *)

(** A [bag] (or [multiset]) is like a set, except that each element
    can appear multiple times rather than just once.  One possible
    implementation is to represent a bag of numbers as a list. *)

Definition bag := natlist.

(** **** Exercise: 3 stars, recommended (bag_functions)  *)
(** Complete the following definitions for the functions
    [count], [sum], [add], and [member] for bags. *)

Fixpoint count (v:nat) (s:bag) : nat:=
  match s with
  | [] => 0
  | x::xs => match beq_nat x v with
             | true => 1 + count v xs
             | false => count v xs
             end
  end.

Example test_count1:              count 1 [1;2;3;1;4;1] = 3.
Proof. reflexivity. Qed.
Example test_count2:              count 6 [1;2;3;1;4;1] = 0.
Proof. reflexivity. Qed.

(** Multiset [sum] is similar to set [union]: [sum a b] contains all
    the elements of [a] and of [b].  (Mathematicians usually define
    [union] on multisets a little bit differently -- using max instead
    of sum -- which is why we don't use that name for this operation.)
    For [sum] we're giving you a header that does not give explicit
    names to the arguments.  Moreover, it uses the keyword
    [Definition] instead of [Fixpoint], so even if you had names for
    the arguments, you wouldn't be able to process them recursively.
    The point of stating the question this way is to encourage you to
    think about whether [sum] can be implemented in another way --
    perhaps by using functions that have already been defined.  *)

Definition sum : bag -> bag -> bag:=app.

Example test_sum1:              count 1 (sum [1;2;3] [1;4;1]) = 3.
Proof. reflexivity. Qed.
(* GRADE_THEOREM 0.5: NatList.test_sum1 *)

Definition add (v:nat) (s:bag) : bag:= v::s.

Example test_add1:                count 1 (add 1 [1;4;1]) = 3.
Proof. reflexivity. Qed.

Example test_add2:                count 5 (add 1 [1;4;1]) = 0.
(* FILL IN HERE *)
Proof. reflexivity. Qed.

(* GRADE_THEOREM 0.5: NatList.test_add1 *)
(* GRADE_THEOREM 0.5: NatList.test_add2 *)

Definition member (v:nat) (s:bag) : bool:=
  match count v s with
  | O => false
  | S n => true
  end.

Example test_member1:             member 1 [1;4;1] = true.
Proof. reflexivity. Qed.

(* GRADE_THEOREM 0.5: NatList.test_member1 *)
(* GRADE_THEOREM 0.5: NatList.test_member2 *)

Example test_member2:             member 2 [1;4;1] = false.
Proof. reflexivity. Qed.

(** [] *)

(** **** Exercise: 3 stars, optional (bag_more_functions)  *)
(** Here are some more [bag] functions for you to practice with. *)

(** When [remove_one] is applied to a bag without the number to remove,
   it should return the same bag unchanged. *)

Fixpoint remove_one (v:nat) (s:bag) : bag:=
  match s with
  | [] => []
  | x::xs => match beq_nat x v with
             | true => xs
             | false => x:: remove_one v xs
             end
  end.

Example test_remove_one1:
  count 5 (remove_one 5 [2;1;5;4;1]) = 0.
Proof. reflexivity. Qed.
Example test_remove_one2:
  count 5 (remove_one 5 [2;1;4;1]) = 0.
Proof. reflexivity. Qed.

Example test_remove_one3:
  count 4 (remove_one 5 [2;1;4;5;1;4]) = 2.
Proof. reflexivity. Qed.
Example test_remove_one4:
  count 5 (remove_one 5 [2;1;5;4;5;1;4]) = 1.
Proof. reflexivity. Qed.

Fixpoint remove_all (v:nat) (s:bag) : bag:=
  match s with
  | [] => []
  | x::xs => match beq_nat x v with
             | true =>  remove_all v xs
             | false => x:: remove_all v xs
             end
  end.

Example test_remove_all1:  count 5 (remove_all 5 [2;1;5;4;1]) = 0.
Proof. reflexivity. Qed.

Example test_remove_all2:  count 5 (remove_all 5 [2;1;4;1]) = 0.
Proof. reflexivity. Qed.

Example test_remove_all3:  count 4 (remove_all 5 [2;1;4;5;1;4]) = 2.
Proof. reflexivity. Qed.

Example test_remove_all4:  count 5 (remove_all 5 [2;1;5;4;5;1;4;5;1;4]) = 0.
Proof. reflexivity. Qed.
Compute member 2 [2;4;1].
Fixpoint subset (s1:bag) (s2:bag) : bool:=
  match s1 with
  | [] => true
  | x::[] => member x s2
  | x::xs => match member x s2 with
             | true => subset xs (remove_one x s2)
             | false => subset xs s2
             end
  end.
Example test_subset1:              subset [1;2] [2;1;4;1] = true.
Proof. reflexivity. Qed.

Example test_subset2:              subset [1;2;2] [2;1;4;1] = false.
Proof. simpl. reflexivity. Qed.

(** [] *)

(** **** Exercise: 3 stars, recommended (bag_theorem)  *)
(** Write down an interesting theorem [bag_theorem] about bags
    involving the functions [count] and [add], and prove it.  Note
    that, since this problem is somewhat open-ended, it's possible
    that you may come up with a theorem which is true, but whose proof
    requires techniques you haven't learned yet.  Feel free to ask for
    help if you get stuck! *)

(*
Theorem bag_theorem : ...
Proof.
  ...
Qed.
*)

(** [] *)

(* ################################################################# *)
(** * Reasoning About Lists *)

(** As with numbers, simple facts about list-processing
    functions can sometimes be proved entirely by simplification.  For
    example, the simplification performed by [reflexivity] is enough
    for this theorem... *)

Theorem nil_app : forall l:natlist,
  [] ++ l = l.
Proof. reflexivity. Qed.

(** ...because the [[]] is substituted into the
    "scrutinee" (the expression whose value is being "scrutinized" by
    the match) in the definition of [app], allowing the match itself
    to be simplified. *)

(** Also, as with numbers, it is sometimes helpful to perform case
    analysis on the possible shapes (empty or non-empty) of an unknown
    list. *)

Theorem tl_length_pred : forall l:natlist,
  pred (length l) = length (tl l).
Proof.
  intros l. destruct l as [| n l'].
  - (* l = nil *)
    reflexivity.
  - (* l = cons n l' *)
    reflexivity.  Qed.

(** Here, the [nil] case works because we've chosen to define
    [tl nil = nil]. Notice that the [as] annotation on the [destruct]
    tactic here introduces two names, [n] and [l'], corresponding to
    the fact that the [cons] constructor for lists takes two
    arguments (the head and tail of the list it is constructing). *)

(** Usually, though, interesting theorems about lists require
    induction for their proofs. *)

(* ----------------------------------------------------------------- *)
(** *** Micro-Sermon *)

(** Simply reading example proof scripts will not get you very far!
    It is important to work through the details of each one, using Coq
    and thinking about what each step achieves.  Otherwise it is more
    or less guaranteed that the exercises will make no sense when you
    get to them.  'Nuff said. *)

(* ================================================================= *)
(** ** Induction on Lists *)

(** Proofs by induction over datatypes like [natlist] are a
    little less familiar than standard natural number induction, but
    the idea is equally simple.  Each [Inductive] declaration defines
    a set of data values that can be built up using the declared
    constructors: a boolean can be either [true] or [false]; a number
    can be either [O] or [S] applied to another number; a list can be
    either [nil] or [cons] applied to a number and a list.

    Moreover, applications of the declared constructors to one another
    are the _only_ possible shapes that elements of an inductively
    defined set can have, and this fact directly gives rise to a way
    of reasoning about inductively defined sets: a number is either
    [O] or else it is [S] applied to some _smaller_ number; a list is
    either [nil] or else it is [cons] applied to some number and some
    _smaller_ list; etc. So, if we have in mind some proposition [P]
    that mentions a list [l] and we want to argue that [P] holds for
    _all_ lists, we can reason as follows:

      - First, show that [P] is true of [l] when [l] is [nil].

      - Then show that [P] is true of [l] when [l] is [cons n l'] for
        some number [n] and some smaller list [l'], assuming that [P]
        is true for [l'].

    Since larger lists can only be built up from smaller ones,
    eventually reaching [nil], these two arguments together establish
    the truth of [P] for all lists [l].  Here's a concrete example: *)

Theorem app_assoc : forall l1 l2 l3 : natlist,
  (l1 ++ l2) ++ l3 = l1 ++ (l2 ++ l3).
Proof.
  intros.
  induction l1 as [| hd tl IHl1'].
  - simpl. reflexivity.
  -  simpl. rewrite IHl1'.
     reflexivity.
Qed.
     (** Notice that, as when doing induction on natural numbers, the
    [as...] clause provided to the [induction] tactic gives a name to
    the induction hypothesis corresponding to the smaller list [l1']
    in the [cons] case. Once again, this Coq proof is not especially
    illuminating as a static written document -- it is easy to see
    what's going on if you are reading the proof in an interactive Coq
    session and you can see the current goal and context at each
    point, but this state is not visible in the written-down parts of
    the Coq proof.  So a natural-language proof -- one written for
    human readers -- will need to include more explicit signposts; in
    particular, it will help the reader stay oriented if we remind
    them exactly what the induction hypothesis is in the second
    case. *)

(** For comparison, here is an informal proof of the same theorem. *)

(** _Theorem_: For all lists [l1], [l2], and [l3],
   [(l1 ++ l2) ++ l3 = l1 ++ (l2 ++ l3)].

   _Proof_: By induction on [l1].

   - First, suppose [l1 = []].  We must show

       ([] ++ l2) ++ l3 = [] ++ (l2 ++ l3),

     which follows directly from the definition of [++].

   - Next, suppose [l1 = n::l1'], with

       (l1' ++ l2) ++ l3 = l1' ++ (l2 ++ l3)

     (the induction hypothesis). We must show

       ((n :: l1') ++ l2) ++ l3 = (n :: l1') ++ (l2 ++ l3).

     By the definition of [++], this follows from

       n :: ((l1' ++ l2) ++ l3) = n :: (l1' ++ (l2 ++ l3)),

     which is immediate from the induction hypothesis.  [] *)

(* ----------------------------------------------------------------- *)
(** *** Reversing a List *)

(** For a slightly more involved example of inductive proof over
    lists, suppose we use [app] to define a list-reversing function
    [rev]: *)

Fixpoint rev (l:natlist) : natlist :=
  match l with
  | nil    => nil
  | h :: t => rev t ++ [h]
  end.

Example test_rev1:            rev [1;2;3] = [3;2;1].
Proof. reflexivity.  Qed.
Example test_rev2:            rev nil = nil.
Proof. reflexivity.  Qed.

(* ----------------------------------------------------------------- *)
(** *** Properties of [rev] *)

(** Now let's prove some theorems about our newly defined [rev].
    For something a bit more challenging than what we've seen, let's
    prove that reversing a list does not change its length.  Our first
    attempt gets stuck in the successor case... *)

Theorem rev_length_firsttry : forall l : natlist,
  length (rev l) = length l.
Proof.
  intros l. induction l as [| n l' IHl'].
  - (* l = [] *)
    reflexivity.
  - (* l = n :: l' *)
    (* This is the tricky case.  Let's begin as usual
       by simplifying. *)
    simpl.
    (* Now we seem to be stuck: the goal is an equality
       involving [++], but we don't have any useful equations
       in either the immediate context or in the global
       environment!  We can make a little progress by using
       the IH to rewrite the goal... *)
    rewrite <- IHl'.
    (* ... but now we can't go any further. *)
Abort.

(** So let's take the equation relating [++] and [length] that
    would have enabled us to make progress and prove it as a separate
    lemma. *)

Theorem app_length : forall l1 l2 : natlist,
  length (l1 ++ l2) = (length l1) + (length l2).
Proof.
  (* WORKED IN CLASS *)
  intros l1 l2. induction l1 as [| n l1' IHl1'].
  - (* l1 = nil *)
    reflexivity.
  - (* l1 = cons *)
    simpl. rewrite -> IHl1'. reflexivity.  Qed.

(** Note that, to make the lemma as general as possible, we
    quantify over _all_ [natlist]s, not just those that result from an
    application of [rev].  This should seem natural, because the truth
    of the goal clearly doesn't depend on the list having been
    reversed.  Moreover, it is easier to prove the more general
    property. *)

(** Now we can complete the original proof. *)

Theorem rev_length : forall l : natlist,
  length (rev l) = length l.
Proof.
  intros l. induction l as [| n l' IHl'].
  - (* l = nil *)
    reflexivity.
  - (* l = cons *)
    simpl. rewrite -> app_length, plus_comm.
    simpl. rewrite -> IHl'. reflexivity.  Qed.

(** For comparison, here are informal proofs of these two theorems:

    _Theorem_: For all lists [l1] and [l2],
       [length (l1 ++ l2) = length l1 + length l2].

    _Proof_: By induction on [l1].

    - First, suppose [l1 = []].  We must show

        length ([] ++ l2) = length [] + length l2,

      which follows directly from the definitions of
      [length] and [++].

    - Next, suppose [l1 = n::l1'], with

        length (l1' ++ l2) = length l1' + length l2.

      We must show

        length ((n::l1') ++ l2) = length (n::l1') + length l2).

      This follows directly from the definitions of [length] and [++]
      together with the induction hypothesis. [] *)

(** _Theorem_: For all lists [l], [length (rev l) = length l].

    _Proof_: By induction on [l].

      - First, suppose [l = []].  We must show

          length (rev []) = length [],

        which follows directly from the definitions of [length]
        and [rev].

      - Next, suppose [l = n::l'], with

          length (rev l') = length l'.

        We must show

          length (rev (n :: l')) = length (n :: l').

        By the definition of [rev], this follows from

          length ((rev l') ++ [n]) = S (length l')

        which, by the previous lemma, is the same as

          length (rev l') + length [n] = S (length l').

        This follows directly from the induction hypothesis and the
        definition of [length]. [] *)

(** The style of these proofs is rather longwinded and pedantic.
    After the first few, we might find it easier to follow proofs that
    give fewer details (which can easily work out in our own minds or
    on scratch paper if necessary) and just highlight the non-obvious
    steps.  In this more compressed style, the above proof might look
    like this: *)

(** _Theorem_:
     For all lists [l], [length (rev l) = length l].

    _Proof_: First, observe that [length (l ++ [n]) = S (length l)]
     for any [l] (this follows by a straightforward induction on [l]).
     The main property again follows by induction on [l], using the
     observation together with the induction hypothesis in the case
     where [l = n'::l']. [] *)

(** Which style is preferable in a given situation depends on
    the sophistication of the expected audience and how similar the
    proof at hand is to ones that the audience will already be
    familiar with.  The more pedantic style is a good default for our
    present purposes. *)

(* ================================================================= *)
(** ** [Search] *)

(** We've seen that proofs can make use of other theorems we've
    already proved, e.g., using [rewrite].  But in order to refer to a
    theorem, we need to know its name!  Indeed, it is often hard even
    to remember what theorems have been proven, much less what they
    are called.

    Coq's [Search] command is quite helpful with this.  Typing
    [Search foo] will cause Coq to display a list of all theorems
    involving [foo].  For example, try uncommenting the following line
    to see a list of theorems that we have proved about [rev]: *)

(*  Search rev. *)

(** Keep [Search] in mind as you do the following exercises and
    throughout the rest of the book; it can save you a lot of time!

    If you are using ProofGeneral, you can run [Search] with [C-c
    C-a C-a]. Pasting its response into your buffer can be
    accomplished with [C-c C-;]. *)

(* ================================================================= *)
(** ** List Exercises, Part 1 *)

(** **** Exercise: 3 stars (list_exercises)  *)
(** More practice with lists: *)

Theorem app_nil_r : forall l : natlist,
  l ++ [] = l.
Proof.
  intros.
  induction l as [| hd tl IHl'].
  - reflexivity.
  - simpl. rewrite  -> IHl'. simpl.
    reflexivity.
Qed.

Theorem rev_app_distr: forall l1 l2 : natlist,
  rev (l1 ++ l2) = rev l2 ++ rev l1.
Proof.
  intros. induction l1 as [| hd tl IHl'].
  - simpl. rewrite -> app_nil_r. reflexivity.
  - simpl. rewrite -> IHl'. rewrite -> app_assoc. reflexivity.
    Qed.

 Theorem rev_involutive : forall l : natlist,
  rev (rev l) = l.
Proof.
  intros.
  induction l.
  - reflexivity.
  - simpl. rewrite rev_app_distr. rewrite IHl.
    reflexivity.
Qed.
    (** There is a short solution to the next one.  If you find yourself
    getting tangled up, step back and try to look for a simpler
    way. *)

Theorem app_assoc4 : forall l1 l2 l3 l4 : natlist,
  l1 ++ (l2 ++ (l3 ++ l4)) = ((l1 ++ l2) ++ l3) ++ l4.
Proof.
intros.
induction l1.
- rewrite app_assoc. reflexivity.
-simpl. rewrite IHl1. reflexivity.
 Qed.
  (** An exercise about your implementation of [nonzeros]: *)

Lemma nonzeros_app : forall l1 l2 : natlist,
  nonzeros (l1 ++ l2) = (nonzeros l1) ++ (nonzeros l2).
Proof.
  intros.
  induction l1.
  - reflexivity.
  - simpl. destruct n.
    + rewrite IHl1. reflexivity.
    + simpl. rewrite IHl1. reflexivity.
Qed.
  (** **** Exercise: 2 stars (beq_natlist)  *)
(** Fill in the definition of [beq_natlist], which compares
    lists of numbers for equality.  Prove that [beq_natlist l l]
    yields [true] for every list [l]. *)

Fixpoint beq_natlist (l1 l2 : natlist) : bool:=
  match l1 with
  | [] => match l2 with
          | [] => true
          | x::xs => false
          end
  | x::xs =>match l2 with
            | [] => false
            | y::ys => match beq_nat x y with
                       | true => andb true (beq_natlist xs ys)
                       | false => false
                       end
            end
  end.

Example test_beq_natlist1 :
  (beq_natlist nil nil = true).
Proof. reflexivity. Qed.
Example test_beq_natlist2 :
  beq_natlist [1;2;3] [1;2;3] = true.
Proof. reflexivity. Qed.
Example test_beq_natlist3 :
  beq_natlist [1;2;3] [1;2;4] = false.
Proof. reflexivity. Qed.

Theorem beq_natlist_refl : forall l:natlist,
  true = beq_natlist l l.
Proof.
  intros.
  induction l as [| hd tl IHl'].
  - reflexivity.
  - simpl.
    rewrite <- beq_nat_refl.
    rewrite IHl'.
    reflexivity.
    Qed.

(* ================================================================= *)
(** ** List Exercises, Part 2 *)

(** Here are a couple of little theorems to prove about your
    definitions about bags above. *)

(** **** Exercise: 1 star (count_member_nonzero)  *)
Theorem count_member_nonzero : forall (s : bag),
  leb 1 (count 1 (1 :: s)) = true.
Proof.
  intros.
  simpl.
  reflexivity.
Qed.

(** The following lemma about [leb] might help you in the next exercise. *)

Theorem ble_n_Sn : forall n,
  leb n (S n) = true.
Proof.
  intros n. induction n as [| n' IHn'].
  - (* 0 *)
    simpl.  reflexivity.
  - (* S n' *)
    simpl.  rewrite IHn'.  reflexivity.  Qed.

(** **** Exercise: 3 stars, advanced (remove_decreases_count)  *)
Theorem remove_decreases_count: forall (s : bag),
  leb (count 0 (remove_one 0 s)) (count 0 s) = true.
Proof. intros.
induction s.
- reflexivity.
- simpl.  destruct n.
  + simpl. rewrite ble_n_Sn. reflexivity.
  + simpl. rewrite IHs. reflexivity.
Qed.

  (** **** Exercise: 3 stars, optional (bag_count_sum)  *)
(** Write down an interesting theorem [bag_count_sum] about bags
    involving the functions [count] and [sum], and prove it using
    Coq.  (You may find that the difficulty of the proof depends on
    how you defined [count]!) *)
(* FILL IN HERE *)
(** [] *)

(** **** Exercise: 4 stars, advanced (rev_injective)  *)
(** Prove that the [rev] function is injective -- that is,

    forall (l1 l2 : natlist), rev l1 = rev l2 -> l1 = l2.

    (There is a hard way and an easy way to do this.) *)

(* FILL IN HERE *)
(** [] *)
Theorem rev_injective: forall (l1 l2 : natlist), rev l1 = rev l2 -> l1 = l2.
Proof.
   intros.
  rewrite <- rev_involutive.
  rewrite <- H.
  rewrite rev_involutive.
  reflexivity.
Qed.

  (* ################################################################# *)
(** * Options *)

(** Suppose we want to write a function that returns the [n]th
    element of some list.  If we give it type [nat -> natlist -> nat],
    then we'll have to choose some number to return when the list is
    too short... *)

Fixpoint nth_bad (l:natlist) (n:nat) : nat :=
  match l with
  | nil => 42  (* arbitrary! *)
  | a :: l' => match beq_nat n O with
               | true => a
               | false => nth_bad l' (pred n)
               end
  end.

(** This solution is not so good: If [nth_bad] returns [42], we
    can't tell whether that value actually appears on the input
    without further processing. A better alternative is to change the
    return type of [nth_bad] to include an error value as a possible
    outcome. We call this type [natoption]. *)

Inductive natoption : Type :=
  | Some : nat -> natoption
  | None : natoption.

(** We can then change the above definition of [nth_bad] to
    return [None] when the list is too short and [Some a] when the
    list has enough members and [a] appears at position [n]. We call
    this new function [nth_error] to indicate that it may result in an
    error. *)

Fixpoint nth_error (l:natlist) (n:nat) : natoption :=
  match l with
  | nil => None
  | a :: l' => match beq_nat n O with
               | true => Some a
               | false => nth_error l' (pred n)
               end
  end.

Example test_nth_error1 : nth_error [4;5;6;7] 0 = Some 4.
Proof. reflexivity. Qed.
Example test_nth_error2 : nth_error [4;5;6;7] 3 = Some 7.
Proof. reflexivity. Qed.
Example test_nth_error3 : nth_error [4;5;6;7] 9 = None.
Proof. reflexivity. Qed.

(** (In the HTML version, the boilerplate proofs of these
    examples are elided.  Click on a box if you want to see one.)

    This example is also an opportunity to introduce one more small
    feature of Coq's programming language: conditional
    expressions... *)

Fixpoint nth_error' (l:natlist) (n:nat) : natoption :=
  match l with
  | nil => None
  | a :: l' => if beq_nat n O then Some a
               else nth_error' l' (pred n)
  end.

(** Coq's conditionals are exactly like those found in any other
    language, with one small generalization.  Since the boolean type
    is not built in, Coq actually supports conditional expressions over
    _any_ inductively defined type with exactly two constructors.  The
    guard is considered true if it evaluates to the first constructor
    in the [Inductive] definition and false if it evaluates to the
    second. *)

(** The function below pulls the [nat] out of a [natoption], returning
    a supplied default in the [None] case. *)

Definition option_elim (d : nat) (o : natoption) : nat :=
  match o with
  | Some n' => n'
  | None => d
  end.

(** **** Exercise: 2 stars (hd_error)  *)
(** Using the same idea, fix the [hd] function from earlier so we don't
    have to pass a default element for the [nil] case.  *)

Definition hd_error (l : natlist) : natoption:=
  match l with | [] => None | x::xs => Some(x) end.
Example test_hd_error1 : hd_error [] = None.
Proof. reflexivity. Qed.
Example test_hd_error2 : hd_error [1] = Some 1.
Proof. reflexivity. Qed.
Example test_hd_error3 : hd_error [5;6] = Some 5.
Proof. reflexivity. Qed.

(** **** Exercise: 1 star, optional (option_elim_hd)  *)
(** This exercise relates your new [hd_error] to the old [hd]. *)

Theorem option_elim_hd : forall (l:natlist) (default:nat),
  hd default l = option_elim default (hd_error l).
Proof.
intros.
destruct l.
- reflexivity.
- reflexivity.
Qed.
End NatList.

(* ################################################################# *)
(** * Partial Maps *)

(** As a final illustration of how data structures can be defined in
    Coq, here is a simple _partial map_ data type, analogous to the
    map or dictionary data structures found in most programming
    languages. *)

(** First, we define a new inductive datatype [id] to serve as the
    "keys" of our partial maps. *)

Inductive id : Type :=
  | Id : nat -> id.

(** Internally, an [id] is just a number.  Introducing a separate type
    by wrapping each nat with the tag [Id] makes definitions more
    readable and gives us the flexibility to change representations
    later if we wish. *)

(** We'll also need an equality test for [id]s: *)

Definition beq_id (x1 x2 : id) :=
  match x1, x2 with
  | Id n1, Id n2 => beq_nat n1 n2
  end.

(** **** Exercise: 1 star (beq_id_refl)  *)
Theorem beq_id_refl : forall x, true = beq_id x x.
Proof.
intros.
destruct x.
simpl.
rewrite <- beq_nat_refl.
reflexivity.
Qed.

(** Now we define the type of partial maps: *)

Module PartialMap.
Export NatList.

Inductive partial_map : Type :=
  | empty  : partial_map
  | record : id -> nat -> partial_map -> partial_map.

(** This declaration can be read: "There are two ways to construct a
    [partial_map]: either using the constructor [empty] to represent an
    empty partial map, or by applying the constructor [record] to
    a key, a value, and an existing [partial_map] to construct a
    [partial_map] with an additional key-to-value mapping." *)

(** The [update] function overrides the entry for a given key in a
    partial map (or adds a new entry if the given key is not already
    present). *)

Definition update (d : partial_map)
                  (x : id) (value : nat)
                  : partial_map :=
  record x value d.

(** Last, the [find] function searches a [partial_map] for a given
    key.  It returns [None] if the key was not found and [Some val] if
    the key was associated with [val]. If the same key is mapped to
    multiple values, [find] will return the first one it
    encounters. *)

Fixpoint find (x : id) (d : partial_map) : natoption :=
  match d with
  | empty         => None
  | record y v d' => if beq_id x y
                     then Some v
                     else find x d'
  end.

(** **** Exercise: 1 star (update_eq)  *)
Theorem update_eq :
  forall (d : partial_map) (x : id) (v: nat),
    find x (update d x v) = Some v.
Proof.
  intros.
  simpl.
  rewrite <- beq_id_refl.
  reflexivity.
  Qed.

(** **** Exercise: 1 star (update_neq)  *)
Theorem update_neq :
  forall (d : partial_map) (x y : id) (o: nat),
    beq_id x y = false -> find x (update d y o) = find x d.
Proof.
  intros.
  simpl.
  rewrite H.
  reflexivity.
  Qed.
End PartialMap.

(** **** Exercise: 2 stars (baz_num_elts)  *)
(** Consider the following inductive definition: *)

Inductive baz : Type :=
  | Baz1 : baz -> baz
  | Baz2 : baz -> bool -> baz.

(** How _many_ elements does the type [baz] have?
    (Explain your answer in words, preferrably English.)
None, you can't construct any, because there is no base case.
*)

(* FILL IN HERE *)
(** [] *)


Require Import Arith Omega.

Definition Even (n:nat) : Prop :=
 exists p:nat, n = 2*p.

Definition Odd (n:nat) : Prop :=
 exists p:nat, n = S (2*p).

 Lemma Even_not_Odd : forall n, Even n -> ~ Odd n.
 Proof.
  intros n [p Hp] [q Hq]; rewrite Hp in Hq.
  omega.
 Qed.

(** Specification of a boolean test for even-ness
*)

Inductive even_spec (n:nat) : bool -> Prop :=
| even_true : forall (Heven : Even n), even_spec n true
| even_false : forall (Hodd : Odd n), even_spec n false.

Definition even_test_ok (f : nat -> bool) :=
 forall n:nat, even_spec n (f n).

Fixpoint even_bool (n:nat) :=
 match n with
   | 0 => true
   | 1 => false
   | S (S p)=> even_bool p
end.

(** For proving even_bool's correctness, we shall use an induction principle
    adaptated to its control structure *)

Lemma nat_double_rect : forall (P:nat->Type),
     P 0 -> P 1 -> (forall p, P p -> P (S (S p))) ->
    forall n:nat, P n.
Proof.
 intros P H0 H1 HS.
 assert (X : forall n, (P n *  P (S n)%type)).
 -  induction n as  [| p IHp]; [auto |  destruct IHp;auto].
 - intro n; now destruct (X n).
Qed.

Lemma even_spec_SS : forall n b, even_spec n b -> even_spec (S (S n)) b.
Proof.
  intros n p [[q Hq] | [q Hq]]; constructor;exists (S q);omega.
Qed.

Lemma even_bool_correct : even_test_ok  even_bool.
Proof.
 intro n;induction n using nat_double_rect.
  constructor;exists 0;trivial.
  constructor;exists 0;trivial.
  cbn;apply even_spec_SS;assumption.
Qed.

(** even_bool_correct can now be used with destruct tactic :

Check even_spec_ind.

even_spec_ind
     : forall (n : nat) (P : bool -> Prop),
       (Even n -> P true) ->
       (Odd n -> P false) -> forall b : bool, even_spec n b -> P b

*)

Lemma even_bool_false (f: nat->bool) (H: even_test_ok f):
 forall n,  Odd n <-> f n = false.
Proof.
 intros n ;destruct (H n).
-  split;try discriminate;intro;destruct (Even_not_Odd n);auto.
- tauto.
Qed.


Set Warnings "-notation-overridden,-parsing".
Require Coq.omega.Omega.
Require Import Coq.omega.Omega.
Require Export Logic.

Inductive ev : nat->Prop :=
  | ev_0 : ev 0
  | ev_SS : forall n : nat, ev n -> ev (S (S n)).

Theorem ev_4: ev 4.
Proof. apply ev_SS. apply ev_SS. apply ev_0.
  Qed.

Theorem ev_4': ev 4.
Proof. apply (ev_SS 2 (ev_SS 0 ev_0)). Qed.

Theorem ev_plus4: forall n, ev n -> ev (4+n).
Proof.
  intros n. simpl. intros Hn.
  apply ev_SS. apply ev_SS. apply Hn.
  Qed.

(* Exercise ev_double *)
Theorem ev_double: forall n,
  ev (double n).
Proof.
  intros n. induction n as [| n' IHn'].
  - simpl. apply ev_0.
  - simpl. apply ev_SS. apply IHn'.
  Qed.

Theorem ev_minus2: forall n,
  ev n -> ev (pred (pred n)).
Proof.
  intros n E.
  inversion E as [| n' E'].
  - simpl. apply ev_0.
  - simpl. apply E'.
  Qed.

Theorem evSS_ev: forall n,
  ev (S (S n)) -> ev n.
Proof.
  intros n E.
  inversion E.
  apply H0.
  Qed.

Theorem one_not_even: ~ev 1.
Proof.
  intros H. inversion H. Qed.

(* Exercise inversion_practice *)
Theorem SSSSev__even: forall n,
  ev (S (S (S (S n)))) -> ev n.
Proof.
  intros n H. inversion H.
  inversion H1.
  apply H3.
  Qed.

Theorem even5_nonsense:
  ev 5 -> 2 + 2 = 9.
Proof.
  intros H. inversion H.
  inversion H1. inversion H3.
  Qed.

Lemma ev_even: forall n,
  ev n -> exists k, n = double k.
Proof.
  intros n E.
  induction E as [| n' E' IH].
  - exists 0. reflexivity.
  - destruct IH as [k' HK'].
  rewrite HK'. exists (S k'). reflexivity.
  Qed.

Theorem ev_even_iff: forall n,
  ev n <-> exists k, n = double k.
Proof.
  intros n. split.
  - apply ev_even.
  - intros [k HK]. rewrite HK.
  apply ev_double.
  Qed.

(* Exercise ev_sum *)
Theorem ev_sum: forall n m,
  ev n -> ev m -> ev (n + m).
Proof.
  intros n m eqn eqm.
  induction eqn as [| n' eqn' IHN].
  - simpl. apply eqm.
  - simpl. apply ev_SS. apply IHN.
  Qed.

(* Exercise ev_alternate *)
Inductive ev' : nat->Prop :=
  | ev'_0 : ev' 0
  | ev'_2 : ev' 2
  | ev'_sum : forall n m,
    ev' n -> ev' m -> ev' (n+m).

Lemma add_two: forall n,
  n + 2 = S (S n).
Proof. intros n. induction n as [| n' IHn'].
  - reflexivity.
  - simpl. rewrite IHn'. reflexivity.
  Qed.

Theorem ev'_ev : forall n,
  ev' n <-> ev n.
Proof.
  intros n. split.
  - intros H.
  induction H as [| |n' m' IHN'].
    + apply ev_0.
    + apply ev_SS. apply ev_0.
    + apply ev_sum.
      ++ apply IHIHN'.
      ++ apply IHev'1.
  - intros H.
  induction H as [| n' IHN].
    + apply ev'_0.
    + rewrite <- add_two.
    apply ev'_sum.
    apply IHIHN.
    apply ev'_2.
  Qed.

(* Exercise ev_ev__ev *)
Theorem ev_ev__ev: forall n m,
  ev (n+m) -> ev n -> ev m.
Proof.
  intros n m eqnm eqn.
  induction eqn as [| n' IHN].
  - simpl in eqnm. apply eqnm.
  - simpl in eqnm. inversion eqnm.
  apply IHIHN in H0.
  apply H0.
  Qed.

Lemma double_distr: forall n m,
  double (n+m) = double n + double m.
Proof.
  intros n m.
  induction n as [|n' IHn'].
  - simpl. reflexivity.
  - simpl. rewrite -> IHn'.
  reflexivity.
  Qed.

Theorem plus_cancel: forall a b c,
  a + (b + c) = a + b + c.
Proof.
  intros a b c.
  rewrite -> plus_assoc.
  reflexivity.
  Qed.

(* Exercise ev_plus_plus *)
Theorem ev_plus_plus: forall n m p,
  ev (n+m) -> ev (n+p) -> ev (m+p).
Proof.
  intros n m p eqnm eqnp.
  assert(H1:ev (n+m+(n+p))).
  { apply ev_sum with (n:=n+m)(m:=n+p).
    - apply eqnm. - apply eqnp. }
  assert(H2:n+m+(n+p)=n+m+n+p).
  { apply plus_cancel with (a:=n+m)(b:=n)(c:=p). }
  rewrite H2 in H1.
  assert(H3:n+m+n+p=n+n+m+p).
  {
    omega. }
  rewrite H3 in H1.
  apply ev_ev__ev with (n:=n+n)(m:=m+p).
  - rewrite <- (plus_assoc (n+n) m p) in H1.
    apply H1.
  - apply ev_even_iff.
    exists n. rewrite -> double_plus.
    reflexivity.
  Qed.

Module Playground.

Inductive le : nat->nat->Prop :=
  | le_n : forall n, le n n
  | le_S : forall n m, (le n m) -> (le n (S m)).

Notation "m <= n" := (le m n).

Theorem test_le1:
  3 <= 3.
Proof.
  apply le_n. Qed.

Theorem test_le2:
  3 <= 6.
Proof.
  apply le_S. apply le_S. apply le_S. apply le_n.
  Qed.

Theorem test_le3:
  (2 <= 1) -> 2 + 2 = 5.
Proof.
  intros H. inversion H. inversion H2.
  Qed.

End Playground.

Definition lt(n m:nat) := le (S n) m.

Notation "m < n" := (lt m n).

Inductive square_of : nat->nat->Prop :=
  | sq : forall n:nat, square_of n (n*n).

Inductive next_nat : nat->nat->Prop :=
  | nn : forall n:nat, next_nat n (S n).

Inductive next_even : nat->nat->Prop :=
  | ne_1 : forall n, ev (S n) -> next_even n (S n)
  | ne_2 : forall n, ev (S (S n)) -> next_even n (S (S n)).

(* Exercise le_exercise *)
Lemma le_trans: forall m n o,
  m<=n -> n<=o -> m<=o.
Proof.
  intros m n o lemn leno.
  induction leno as [| n' o' H].
  - apply lemn.
  - apply le_S. apply H.
  Qed.

Theorem O_le_n : forall n,
  0 <= n.
Proof.
  intros n. induction n as [| n' H].
  - apply le_n.
  - apply le_S in H. apply H.
  Qed.

Theorem n_le_m__Sn_le_Sm : forall n m,
  n<=m -> S n <= S m.
Proof.
  intros n m eq. induction eq.
  - apply le_n.
  - apply le_S. apply IHeq.
  Qed.

Theorem Sn_le_Sm__n_le_m : forall n m,
  S n <= S m -> n <= m.
Proof.
  intros n m eq. inversion eq.
  - apply le_n.
  - apply le_trans with (m:=n)(n:=(S n))(o:=m).
    + apply le_S. reflexivity.
    + apply H0.
  Qed.

Theorem le_plus_l: forall a b,
  a <= a + b.
Proof.
  intros a b.
  induction a as [| a'].
  - simpl. apply O_le_n.
  - simpl. apply n_le_m__Sn_le_Sm.
  apply IHa'.
  Qed.

Theorem plus_lt: forall n1 n2 m,
  n1 + n2 < m ->
  n1 < m /\ n2 < m.
Proof.
  unfold lt.
  intros n1 n2 m eq1. split.
  - assert (n1 <= n1+n2).
  { apply le_plus_l with (a:=n1)(b:=n2). }
  apply n_le_m__Sn_le_Sm in H.
  apply le_trans with (m:=S n1)(n:=S (n1+n2))(o:=m).
  + apply H.
  + apply eq1.
  - assert (n2 <= n2+n1).
  { apply le_plus_l with (a:=n2)(b:=n1). }
  rewrite plus_comm in H.
  apply n_le_m__Sn_le_Sm in H.
  apply le_trans with (m:=S n2)(n:=S (n1+n2))(o:=m).
  + apply H.
  + apply eq1.
  Qed.

Theorem lt_S: forall n m,
  n < m ->
  n < S m.
Proof.
  unfold lt. intros n m eq.
  apply le_S in eq. apply eq.
  Qed.

Theorem leb_complete: forall n m,
  leb n m = true ->
  n <= m.
Proof.
  intros n m eq.
  generalize dependent m.
  induction n as [| n'].
  - intros n eq. apply O_le_n.
  - induction m as [| m'].
    + intros eq. simpl in eq. inversion eq.
    + intros eq. simpl in eq.
    apply IHn' in eq.
    apply n_le_m__Sn_le_Sm in eq.
    apply eq.
  Qed.

Theorem leb_refl : forall n,
  leb n n = true.
Proof.
  induction n as [|n'].
  - simpl. reflexivity.
  - simpl. apply IHn'.
  Qed.

Theorem n_leb_m__n_leb_Sm : forall n m,
  leb n m = true ->
  leb n (S m) = true.
Proof.
  intros n m eq.
  generalize dependent m.
  induction n as [|n'].
  - intros m eq. simpl. reflexivity.
  - intros m eq. simpl. induction m as [|m'].
    + simpl in eq. inversion eq.
    + simpl in eq. apply IHn' in eq.
    apply eq.
  Qed.

Theorem leb_correct : forall n m,
  n <= m ->
  leb n m = true.
Proof.
  intros n m eq.
  generalize dependent n.
  induction m as [|m'].
  - intros n eq. inversion eq. simpl. reflexivity.
  - intros n eq. inversion eq.
    + simpl. apply leb_refl.
    + apply IHm' in H0.
    apply n_leb_m__n_leb_Sm.
    apply H0.
  Qed.

Theorem leb_true_trans : forall n m o,
  leb n m = true ->
  leb m o = true ->
  leb n o = true.
Proof.
  intros n m o eqnm eqmoo.
  apply leb_complete in eqnm.
  apply leb_complete in eqmoo.
  assert(H:n<=o).
  { apply le_trans with (m:=n)(n:=m)(o:=o).
    apply eqnm. apply eqmoo. }
  apply leb_correct in H.
  apply H.
  Qed.

(* Exercise leb_iff *)
Theorem leb_iff : forall n m,
  leb n m = true <-> n <= m.
Proof.
  intros n m. split.
  - apply leb_complete.
  - intros eq. apply leb_correct.
  apply eq.
  Qed.

Module R.

(* Exercise R_provability *)
Inductive R : nat->nat->nat->Prop :=
  | c1: R 0 0 0
  | c2: forall m n o, R m n o -> R (S m) n (S o)
  | c3: forall m n o, R m n o -> R m (S n) (S o)
  | c4: forall m n o, R (S m) (S n) (S (S o)) -> R m n o
  | c5: forall m n o, R m n o -> R n m o.

Example prov1: (R 1 1 2).
Proof.
  apply c2.
  apply c3.
  apply c1.
  Qed.

Example prov2: (R 2 2 6).
Proof.
  Admitted.

(* Exercise R_fact *)
Definition fR:nat->nat->nat :=
  fun a =>
    fun b =>
      a + b.

Theorem R_equiv_fR: forall m n o,
  R m n o <-> fR m n = o.
Proof.
  intros m n o.
  split.
  - intros eqR. unfold fR.
  induction eqR.
    + reflexivity.
    + simpl. rewrite IHeqR. reflexivity.
    + rewrite <- IHeqR. rewrite <- plus_n_Sm.
      reflexivity.
    + simpl in IHeqR. inversion IHeqR.
    rewrite <- plus_n_Sm in H0.
    inversion H0.
    reflexivity.
    + rewrite -> plus_comm in IHeqR.
    apply IHeqR.
  - generalize dependent o.
    unfold fR.
    induction m as [|m IHm'].
    + induction n as [|n IHn'].
      * simpl in *.
        intros o eq. rewrite <- eq. apply c1.
      * intros o eq. simpl in *.
        rewrite <- eq.
        apply c3. apply IHn'. reflexivity.
    + induction n as [|n IHn'].
      * simpl in *.
        intros o eq.
        rewrite -> plus_comm in eq.
        simpl in *. rewrite <- eq.
        apply c2. apply IHm'.
        rewrite -> plus_comm. simpl.
        reflexivity.
      * intros o eq.
        simpl in eq. rewrite <- plus_n_Sm in eq.
        rewrite <-  eq.
        apply c2. apply IHm'.
        rewrite <- plus_n_Sm.
        reflexivity.
  Qed.

End R.

(* Exercise subsequence *)

(* Exercise R_provability2 *)

Inductive R:nat->list nat->Prop :=
  | c1: R 0 []
  | c2: forall n l,
      R n l ->
      R (S n) (n::l)
  | c3: forall n l,
      R (S n) l ->
      R n l.

Example Rlist_prov1: R 2 [1;0].
Proof.
  apply c2. apply c2. apply c1.
  Qed.

Example Rlist_prov2: R 1 [1;2;1;0].
Proof.
  apply c3. apply c2. apply c3. apply c3.
  apply c2. apply c2. apply c2. apply c1.
  Qed.

Example Rlist_prov3: R 6 [3;2;1;0].
Proof.
  Admitted.

Inductive reg_exp{T:Type} : Type :=
  | EmptySet : reg_exp
  | EmptyStr : reg_exp
  | Char : T->reg_exp
  | App: reg_exp->reg_exp->reg_exp
  | Union: reg_exp->reg_exp->reg_exp
  | Star: reg_exp->reg_exp.

 Inductive exp_match {T} : list T -> reg_exp -> Prop :=
| MEmpty : exp_match [] EmptyStr
| MChar : forall x, exp_match [x] (Char x)
| MApp : forall s1 re1 s2 re2,
           exp_match s1 re1 ->
           exp_match s2 re2 ->
           exp_match (s1 ++ s2) (App re1 re2)
| MUnionL : forall s1 re1 re2,
              exp_match s1 re1 ->
              exp_match s1 (Union re1 re2)
| MUnionR : forall re1 s2 re2,
              exp_match s2 re2 ->
              exp_match s2 (Union re1 re2)
| MStar0 : forall re, exp_match [] (Star re)
| MStarApp : forall s1 s2 re,
               exp_match s1 re ->
               exp_match s2 (Star re) ->
               exp_match (s1 ++ s2) (Star re).

 Notation "s =~ re" := (exp_match s re) (at level 80).

Example reg_exp_ex1: [1] =~ Char 1.
Proof.
  apply MChar.
  Qed.

Example reg_exp_ex2: [1;2] =~ App (Char 1) (Char 2).
Proof.
  apply (MApp [1] _ [2]).
  - apply MChar.
  - apply MChar.
  Qed.

Example reg_exp_ex3: ~([1;2] =~ Char 1).
Proof.
  intros H. inversion H.
  Qed.

Fixpoint reg_exp_of_list{T}(l:list T) :=
  match l with
  | [] => EmptyStr
  | x::l' => App (Char x) (reg_exp_of_list l')
  end.

Example reg_exp_ex4: [1;2;3] =~ reg_exp_of_list [1;2;3].
Proof.
  simpl. apply (MApp [1]).
  { apply MChar. }
  apply (MApp [2]).
  { apply MChar. }
  apply (MApp [3]).
  { apply MChar. }
  apply MEmpty.
  Qed.

Lemma MStar1:
  forall T s (re:@reg_exp T),
  s =~ re ->
  s =~ Star re.
Proof.
  intros T s re H.
  rewrite <- (app_nil_r _ s).
  apply (MStarApp s [] re).
  - apply H.
  - apply MStar0.
  Qed.

(* Exercise exp_match_ex1 *)
Lemma empty_is_empty: forall T (s:list T),
  ~(s =~ EmptySet).
Proof.
  intros T s. unfold not. intros eq.
  inversion eq.
  Qed.

Lemma MUnion': forall T (s:list T)
  (re1 re2 : @reg_exp T),
  s =~ re1 \/ s =~ re2 ->
  s =~ Union re1 re2.
Proof.
  intros T s re1 re2 [H | H].
  - apply MUnionL. apply H.
  - apply MUnionR. apply H.
  Qed.

Lemma MStar': forall T (ss:list (list T))
  (re:reg_exp),
  (forall s, In s ss -> s =~ re) ->
  fold app ss [] =~ Star re.
Proof.
  intros T ss re eq.
  induction ss as [|hh tt IHs'].
  - simpl. apply MStar0.
  - simpl. Admitted.

(* Exercise reg_exp_of_list *)
Lemma reg_exp_of_list_spec: forall T (s1 s2:list T),
  s1 =~ reg_exp_of_list s2 <-> s1 = s2.
Proof.
  intros T s1 s2. split.
  - intros eq. induction s2 as [|hh tt IHs'].
    + simpl in eq. inversion eq. reflexivity.
    + simpl in *. Admitted.

Fixpoint re_chars{T}(re:reg_exp) : list T :=
  match re with
  | EmptySet => []
  | EmptyStr => []
  | Char x => [x]
  | App re1 re2 => re_chars re1 ++ re_chars re2
  | Union re1 re2 => re_chars re1 ++ re_chars re2
  | Star re => re_chars re
  end.

Theorem in_re_match: forall T (s:list T)
  (re:reg_exp)(x:T),
  s =~ re ->
  In x s ->
  In x (re_chars re).
Proof.
  intros T s re x Hmatch Hin.
  induction Hmatch
    as [
        |x'
        |s1 re1 s2 re2 Hmatch1 IH1 Hmatch2 IH2
        |s1 re1 re2 Hmatch IH | re1 s2 re2 Hmatch IH
        |re |s1 s2 re Hmatch1 IH1 Hmatch2 IH2
        ].
  - simpl. apply Hin.
  - simpl. simpl in Hin. apply Hin.
  - simpl. rewrite in_app_iff in *.
    destruct Hin as [Hin | Hin].
    + left. apply (IH1 Hin).
    + right. apply (IH2 Hin).
  - simpl. rewrite in_app_iff.
  left. apply (IH Hin).
  - simpl. rewrite in_app_iff.
  right. apply (IH Hin).
  - simpl. inversion Hin.
  - simpl. apply in_app_iff in Hin.
    destruct Hin as [Hin | Hin].
    + apply (IH1 Hin).
    + apply (IH2 Hin).
  Qed.

(* Exercise re_not_empty *)

Lemma star_app: forall T (s1 s2 : list T)
  (re:@reg_exp T),
  s1 =~ Star re ->
  s2 =~ Star re ->
  s1 ++ s2 =~ Star re.
Proof.
  intros T s1 s2 re H1.
  remember (Star re) as re'.
  generalize dependent s2.
  induction H1.
  - inversion Heqre'.
  - inversion Heqre'.
  - inversion Heqre'.
  - inversion Heqre'.
  - inversion Heqre'.
  - inversion Heqre'.
  intros s2 H. simpl. apply H.
  - inversion Heqre'.
  rewrite H0 in *.
  intros ss2 H1.
  rewrite <- app_assoc.
  apply MStarApp.
  + apply H1_.
  + apply ((IHexp_match2 Heqre') ss2).
  apply H1.
  Qed.

(* Exercise exp_match_ex2 *)

(* Exercise pumping *)

Theorem filter_not_empty_In: forall n l,
  filter (beq_nat n) l <> [] ->
  In n l.
Proof.
  intros n l. induction l as [|m l' IHl'].
  - simpl. intros H. apply H. reflexivity.
  - simpl. destruct (beq_nat n m) eqn:H.
    + intros _.
      rewrite beq_nat_true_iff in H.
      left. symmetry. apply H.
    + intros H'. right. apply IHl'. apply H'.
  Qed.

Module FirstTry.

Inductive reflect: Prop->bool->Prop :=
  | ReflectT: forall(P:Prop), P->reflect P true
  | ReflectF: forall(P:Prop), ~P->reflect P false.

End FirstTry.

Inductive reflect(P:Prop): bool->Prop :=
  | ReflectT: P->reflect P true
  | ReflectF: ~P->reflect P false.

Theorem iff_reflect: forall P b,
  (P <-> b=true) ->
  reflect P b.
Proof.
  intros P b eq. destruct b.
  - apply ReflectT. apply eq. reflexivity.
  - apply ReflectF. unfold not. intros HP.
  apply eq in HP. inversion HP.
  Qed.

(* Exercise reflect_iff *)
Theorem reflect_iff: forall P b,
  reflect P b -> (P <-> b = true).
Proof.
  intros P b eq.
  destruct b.
  - split.
    + intros HP. reflexivity.
    + intros _. inversion eq. apply H.
  - split.
    + intros HP. inversion eq. apply H in HP.
    inversion HP.
    + intros HFalse. inversion HFalse.
  Qed.

Lemma beq_natP: forall n m,
  reflect (n=m) (beq_nat n m).
Proof.
  intros n m. apply iff_reflect.
  rewrite beq_nat_true_iff.
  reflexivity.
  Qed.

Theorem filter_not_empty_In': forall n l,
  filter (beq_nat n) l <> [] ->
  In n l.
Proof.
  intros n l. induction l as [|m l' IHl'].
  - simpl. intros H. apply H. reflexivity.
  - simpl. destruct (beq_natP n m) as [H | H].
    + intros _. rewrite H. left. reflexivity.
    + intros H'. right. apply IHl'. apply H'.
  Qed.

(* Exercise beq_natP_practice *)
Fixpoint count n l :=
  match l with
  | [] => 0
  | m::l' =>
    (if beq_nat n m then 1 else 0) + count n l'
  end.

Theorem beq_natP_practice' : forall n l,
  count n l = 0 ->
  ~(In n l).
Proof. unfold not.
    intros n l eq.
    induction l as [|hh tt IHl'].
    - simpl. intros F. apply F.
    - simpl. intros [H|H].
      + rewrite H in *. apply IHl'.
        simpl in *.
        Search beq_nat.
        rewrite Induction.beq_nat_refl in eq.
        inversion eq.
        simpl in eq.
        rewrite Induction.beq_nat_refl in eq.
        inversion eq.
      + apply IHl'.
        simpl in *.
        destruct (beq_nat n hh).
        * inversion eq.
        * simpl in eq. apply eq.
        * apply H.
  Qed.

Theorem beq_natP_practice : forall n l,
  count n l = 0 ->
  ~(In n l).
Proof.
  unfold not. intros n l. induction l as [|m l' IHl'].
  - simpl. intros _ F. apply F.
  - simpl in *. destruct (beq_natP n m) as [H|H].
    + intros eq [H1|H1].
      * inversion eq.
      * inversion eq.
    + simpl in *. intros eq [H1|H1].
      * rewrite H1 in H. apply H. reflexivity.
      * apply IHl' in eq.
        ** apply eq.
        ** apply H1.
  Qed.

(* Exercise nostutter *)

(* Exercise filter_challenge *)

(* Exercise filter_challenge2 *)

(* Exercise palindromes *)

(* Exercise palindrome_converse *)

(* Exercise NoDup *)

(* Exercise pigeonhole principle *)


(** * Prop: Propositions and Evidence *)

Require Export MoreCoq.

(** In previous chapters, we have seen many examples of factual
    claims (_propositions_) and ways of presenting evidence of their
    truth (_proofs_).  In particular, we have worked extensively with
    _equality propositions_ of the form [e1 = e2], with
    implications ([P -> Q]), and with quantified propositions
    ([forall x, P]).

    This chapter will take us on a first tour of the
    propositional (logical) side of Coq.
    In particular, we will expand our repertoire of primitive
    propositions to include _user-defined_ propositions, not just
    equality propositions (which are more-or-less "built in" to Coq).
*)

(* ##################################################### *)
(** * Inductively Defined Propositions *)

(**  As a running example, let's
    define a simple property of natural numbers -- we'll call it
    "[beautiful]." *)

(** Informally, a number is [beautiful] if it is [0], [3], [5], or the
    sum of two [beautiful] numbers.

    More pedantically, we can define [beautiful] numbers by giving four
    rules:

       - Rule [b_0]: The number [0] is [beautiful].
       - Rule [b_3]: The number [3] is [beautiful].
       - Rule [b_5]: The number [5] is [beautiful].
       - Rule [b_sum]: If [n] and [m] are both [beautiful], then so is
         their sum. *)

(** We will see many definitions like this one during the rest
    of the course, and for purposes of informal discussions, it is
    helpful to have a lightweight notation that makes them easy to
    read and write.  _Inference rules_ are one such notation: *)
(**
                              -----------                               (b_0)
                              beautiful 0

                              ------------                              (b_3)
                              beautiful 3

                              ------------                              (b_5)
                              beautiful 5

                       beautiful n     beautiful m
                       ---------------------------                      (b_sum)
                              beautiful (n+m)
*)

(** Each of the textual rules above is reformatted here as an
    inference rule; the intended reading is that, if the _premises_
    above the line all hold, then the _conclusion_ below the line
    follows.  For example, the rule [b_sum] says that, if [n] and [m]
    are both [beautiful] numbers, then it follows that [n+m] is
    [beautiful] too.  If a rule has no premises above the line, then
    its conclusion hold unconditionally.

    These rules _define_ the property [beautiful].  That is, if we
    want to convince someone that some particular number is [beautiful],
    our argument must be based on these rules.  For a simple example,
    suppose we claim that the number [5] is [beautiful].  To support
    this claim, we just need to point out that rule [b_5] says so.
    Or, if we want to claim that [8] is [beautiful], we can support our
    claim by first observing that [3] and [5] are both [beautiful] (by
    rules [b_3] and [b_5]) and then pointing out that their sum, [8],
    is therefore [beautiful] by rule [b_sum].  This argument can be
    expressed graphically with the following _proof tree_: *)
(**
         ----------- (b_3)   ----------- (b_5)
         beautiful 3         beautiful 5
         ------------------------------- (b_sum)
                   beautiful 8
    Of course, there are other ways of using these rules to argue that
    [8] is [beautiful], for instance:
         ----------- (b_5)   ----------- (b_3)
         beautiful 5         beautiful 3
         ------------------------------- (b_sum)
                   beautiful 8
*)

(** **** Exercise: 1 star (varieties_of_beauty) *)
(** How many different ways are there to show that [8] is [beautiful]? *)

(* Infinitely many*)
(** [] *)

(** In Coq, we can express the definition of [beautiful] as
    follows: *)

Inductive beautiful : nat -> Prop :=
  b_0   : beautiful 0
| b_3   : beautiful 3
| b_5   : beautiful 5
| b_sum : forall n m, beautiful n -> beautiful m -> beautiful (n+m).

(** The first line declares that [beautiful] is a proposition -- or,
    more formally, a family of propositions "indexed by" natural
    numbers.  (That is, for each number [n], the claim that "[n] is
    [beautiful]" is a proposition.)  Such a family of propositions is
    often called a _property_ of numbers.  Each of the remaining lines
    embodies one of the rules for [beautiful] numbers.

    The rules introduced this way have the same status as proven
    theorems; that is, they are true axiomatically.
    So we can use Coq's [apply] tactic with the rule names to prove
    that particular numbers are [beautiful].  *)

Theorem three_is_beautiful: beautiful 3.
Proof.
   (* This simply follows from the rule [b_3]. *)
   apply b_3.
Qed.

Theorem eight_is_beautiful: beautiful 8.
Proof.
   (* First we use the rule [b_sum], telling Coq how to
      instantiate [n] and [m]. *)
   apply b_sum with (n:=3) (m:=5).
   (* To solve the subgoals generated by [b_sum], we must provide
      evidence of [beautiful 3] and [beautiful 5]. Fortunately we
      have rules for both. *)
   apply b_3.
   apply b_5.
Qed.

(** As you would expect, we can also prove theorems that have
hypotheses about [beautiful]. *)

Theorem beautiful_plus_eight: forall n, beautiful n -> beautiful (8+n).
Proof.
  intros n B.
  apply b_sum with (n:=8) (m:=n).
  apply eight_is_beautiful.
  apply B.
Qed.

(** **** Exercise: 2 stars (b_timesm) *)
Theorem b_timesm: forall n m, beautiful n -> beautiful (m*n).
Proof.
  intros.
  induction m as [|m']. simpl. apply b_0.
  simpl. apply b_sum with (n := n) (m := m' * n).
  apply H. apply IHm'.
  Qed.

(* ####################################################### *)
(** ** Induction Over Evidence *)

(** Besides _constructing_ evidence that numbers are beautiful, we can
    also _reason about_ such evidence. *)

(** The fact that we introduced [beautiful] with an [Inductive]
    declaration tells Coq not only that the constructors [b_0], [b_3],
    [b_5] and [b_sum] are ways to build evidence, but also that these
    two constructors are the _only_ ways to build evidence that
    numbers are beautiful. *)

(** In other words, if someone gives us evidence [E] for the assertion
    [beautiful n], then we know that [E] must have one of four shapes:

      - [E] is [b_0] (and [n] is [O]),
      - [E] is [b_3] (and [n] is [3]),
      - [E] is [b_5] (and [n] is [5]), or
      - [E] is [b_sum n1 n2 E1 E2] (and [n] is [n1+n2], where [E1] is
        evidence that [n1] is beautiful and [E2] is evidence that [n2]
        is beautiful). *)

(** This permits us to _analyze_ any hypothesis of the form [beautiful
    n] to see how it was constructed, using the tactics we already
    know.  In particular, we can use the [induction] tactic that we
    have already seen for reasoning about inductively defined _data_
    to reason about inductively defined _evidence_.

    To illustrate this, let's define another property of numbers: *)

Inductive gorgeous : nat -> Prop :=
  g_0 : gorgeous 0
| g_plus3 : forall n, gorgeous n -> gorgeous (3+n)
| g_plus5 : forall n, gorgeous n -> gorgeous (5+n).

(** **** Exercise: 1 star (gorgeous_tree) *)
(** Write out the definition of [gorgeous] numbers using inference rule
    notation.
-----------
gorgeous 0

gorgeous n
---------------
gorgeous(3 + n)

gorgeous n
----------------
gorgeous(5 + n)

*)

(** **** Exercise: 1 star (gorgeous_plus13) *)
Theorem gorgeous_plus13: forall n,
  gorgeous n -> gorgeous (13+n).
Proof.
  intros.
  apply g_plus5 with (n := 8 + n).
  apply g_plus5 with (n := 3 + n).
  apply g_plus3 with (n := n).
  apply H.
  Qed.

(** It seems intuitively obvious that, although [gorgeous] and
    [beautiful] are presented using slightly different rules, they are
    actually the same property in the sense that they are true of the
    same numbers.  Indeed, we can prove this. *)

Theorem gorgeous__beautiful : forall n,
  gorgeous n -> beautiful n.
Proof.
   intros n H.
   induction H as [|n'|n'].
   Case "g_0".
       apply b_0.
   Case "g_plus3".
       apply b_sum. apply b_3.
       apply IHgorgeous.
   Case "g_plus5".
       apply b_sum. apply b_5. apply IHgorgeous.
Qed.

(** Notice that the argument proceeds by induction on the _evidence_ [H]! *)

(** Let's see what happens if we try to prove this by induction on [n]
   instead of induction on the evidence [H]. *)

Theorem gorgeous__beautiful_FAILED : forall n,
  gorgeous n -> beautiful n.
Proof.
   intros. induction n as [| n'].
   Case "n = 0". apply b_0.
   Case "n = S n'". (* We are stuck! *)
Abort.

(** The problem here is that doing induction on [n] doesn't yield a
    useful induction hypothesis. Knowing how the property we are
    interested in behaves on the predecessor of [n] doesn't help us
    prove that it holds for [n]. Instead, we would like to be able to
    have induction hypotheses that mention other numbers, such as [n -
    3] and [n - 5]. This is given precisely by the shape of the
    constructors for [gorgeous]. *)

(** **** Exercise: 2 stars (gorgeous_sum) *)
Theorem gorgeous_sum : forall n m,
  gorgeous n -> gorgeous m -> gorgeous (n + m).
Proof.
  intros.
  induction H.
  simpl. apply H0.
  apply g_plus3 with (n := n+m). apply IHgorgeous.
  apply g_plus5 with (n := n+m). apply IHgorgeous.
  Qed.

(** **** Exercise: 3 stars, advanced (beautiful__gorgeous) *)
Theorem beautiful__gorgeous : forall n, beautiful n -> gorgeous n.
Proof.
  intros.
  induction H.
  apply g_0.
  apply g_plus3 with (n := 0). apply g_0.
  apply g_plus5 with (n := 0). apply g_0.
  apply gorgeous_sum. apply IHbeautiful1. apply IHbeautiful2.
  Qed.

(** **** Exercise: 3 stars, optional (g_times2) *)
(** Prove the [g_times2] theorem below without using [gorgeous__beautiful].
    You might find the following helper lemma useful. *)

Lemma helper_g_times2 : forall x y z, x + (z + y) = z + x + y.
Proof.
  intros.
  rewrite -> plus_swap. apply plus_assoc. Qed.

Theorem g_times2: forall n, gorgeous n -> gorgeous (2*n).
Proof.
   intros n H. simpl.
   induction H.
   simpl. apply g_0.
   apply g_plus3 with (n := n + (3 + n + 0)).
   rewrite -> helper_g_times2. apply g_plus3 with (n := n + n + 0).
   rewrite <- plus_assoc. apply IHgorgeous.
   apply g_plus5 with (n := n + (5 + n + 0)).
   rewrite -> helper_g_times2. apply g_plus5 with (n := n + n + 0).
   rewrite <- plus_assoc. apply IHgorgeous.
   Qed.

(* ####################################################### *)
(** ** From Boolean Functions to Propositions *)

(** In chapter [Basics] we defined a _function_ [evenb] that tests a
    number for evenness, yielding [true] if so.  We can use this
    function to define the _proposition_ that some number [n] is
    even: *)

Definition even (n:nat) : Prop :=
  evenb n = true.

(** That is, we can define "[n] is even" to mean "the function [evenb]
    returns [true] when applied to [n]."

    Note that here we have given a name
    to a proposition using a [Definition], just as we have
    given names to expressions of other sorts. This isn't a fundamentally
    new kind of proposition;  it is still just an equality. *)

(** Another alternative is to define the concept of evenness
    directly.  Instead of going via the [evenb] function ("a number is
    even if a certain computation yields [true]"), we can say what the
    concept of evenness means by giving two different ways of
    presenting _evidence_ that a number is even. *)

Inductive ev : nat -> Prop :=
  | ev_0 : ev O
  | ev_SS : forall n:nat, ev n -> ev (S (S n)).

(** This definition says that there are two ways to give
    evidence that a number [m] is even.  First, [0] is even, and
    [ev_0] is evidence for this.  Second, if [m = S (S n)] for some
    [n] and we can give evidence [e] that [n] is even, then [m] is
    also even, and [ev_SS n e] is the evidence. *)

(** **** Exercise: 1 star (double_even) *)

Theorem double_even : forall n,
  ev (double n).
Proof.
  intros.
  induction n as [|n'].
  simpl. apply ev_0.
  simpl. apply ev_SS. apply IHn'.
  Qed.

(** *** Discussion: Computational vs. Inductive Definitions *)

(** We have seen that the proposition "[n] is even" can be
    phrased in two different ways -- indirectly, via a boolean testing
    function [evenb], or directly, by inductively describing what
    constitutes evidence for evenness.  These two ways of defining
    evenness are about equally easy to state and work with.  Which we
    choose is basically a question of taste.

    However, for many other properties of interest, the direct
    inductive definition is preferable, since writing a testing
    function may be awkward or even impossible.

    One such property is [beautiful].  This is a perfectly sensible
    definition of a set of numbers, but we cannot translate its
    definition directly into a Coq Fixpoint (or into a recursive
    function in any other common programming language).  We might be
    able to find a clever way of testing this property using a
    [Fixpoint] (indeed, it is not too hard to find one in this case),
    but in general this could require arbitrarily deep thinking.  In
    fact, if the property we are interested in is uncomputable, then
    we cannot define it as a [Fixpoint] no matter how hard we try,
    because Coq requires that all [Fixpoint]s correspond to
    terminating computations.

    On the other hand, writing an inductive definition of what it
    means to give evidence for the property [beautiful] is
    straightforward. *)

(** **** Exercise: 1 star (ev__even) *)
(** Here is a proof that the inductive definition of evenness implies
    the computational one. *)

Theorem ev__even : forall n,
  ev n -> even n.
Proof.
  intros n E. induction E as [| n' E'].
  Case "E = ev_0".
    unfold even. reflexivity.
  Case "E = ev_SS n' E'".
    unfold even. apply IHE'.
Qed.

(** Could this proof also be carried out by induction on [n] instead
    of [E]?  If not, why not? *)

(* No, for the same reason we couldn't with gorgeous__beautiful, we need
** an induction hypothesis where each inductive step is concerned with
** n-2, not n-1 *)
(** [] *)

(** The induction principle for inductively defined propositions does
    not follow quite the same form as that of inductively defined
    sets.  For now, you can take the intuitive view that induction on
    evidence [ev n] is similar to induction on [n], but restricts our
    attention to only those numbers for which evidence [ev n] could be
    generated.  We'll look at the induction principle of [ev] in more
    depth below, to explain what's really going on. *)

(** **** Exercise: 1 star (l_fails) *)
(** The following proof attempt will not succeed.
     Theorem l : forall n,
       ev n.
     Proof.
       intros n. induction n.
         Case "O". simpl. apply ev_0.
         Case "S".
           ...
   Intuitively, we expect the proof to fail because not every
   number is even. However, what exactly causes the proof to fail?

(* We have no rule in our inductive definition regarding ev (S n) *)
*)

(** [] *)

(** **** Exercise: 2 stars (ev_sum) *)
(** Here's another exercise requiring induction. *)

Theorem ev_sum : forall n m,
   ev n -> ev m -> ev (n+m).
Proof.
  intros. induction H.
  simpl. apply H0.
  simpl. apply ev_SS. apply IHev.
  Qed.

(* ####################################################### *)
(** ** [Inversion] on Evidence *)

(** Another situation where we want to analyze evidence for evenness
    is when proving that, if [n] is even, then [pred (pred n))] is
    too.  In this case, we don't need to do an inductive proof.  The
    right tactic turns out to be [inversion].  *)

Theorem ev_minus2: forall n,
  ev n -> ev (pred (pred n)).
Proof.
  intros n E.
  inversion E as [| n' E'].
  Case "E = ev_0". simpl. apply ev_0.
  Case "E = ev_SS n' E'". simpl. apply E'.  Qed.

(** **** Exercise: 1 star, optional (ev_minus2_n) *)
(** What happens if we try to use [destruct] on [n] instead of [inversion] on [E]? *)

(* You will get stuck with ev (pred n') *)
(** [] *)

(** Another example, in which [inversion] helps narrow down to
the relevant cases. *)

Theorem SSev__even : forall n,
  ev (S (S n)) -> ev n.
Proof.
  intros n E.
  inversion E as [| n' E'].
  apply E'. Qed.

(** These uses of [inversion] may seem a bit mysterious at first.
    Until now, we've only used [inversion] on equality
    propositions, to utilize injectivity of constructors or to
    discriminate between different constructors.  But we see here
    that [inversion] can also be applied to analyzing evidence
    for inductively defined propositions.

    (You might also expect that [destruct] would be a more suitable
    tactic to use here. Indeed, it is possible to use [destruct], but
    it often throws away useful information, and the [eqn:] qualifier
    doesn't help much in this case.)

    Here's how [inversion] works in general.  Suppose the name
    [I] refers to an assumption [P] in the current context, where
    [P] has been defined by an [Inductive] declaration.  Then,
    for each of the constructors of [P], [inversion I] generates
    a subgoal in which [I] has been replaced by the exact,
    specific conditions under which this constructor could have
    been used to prove [P].  Some of these subgoals will be
    self-contradictory; [inversion] throws these away.  The ones
    that are left represent the cases that must be proved to
    establish the original goal.

    In this particular case, the [inversion] analyzed the construction
    [ev (S (S n))], determined that this could only have been
    constructed using [ev_SS], and generated a new subgoal with the
    arguments of that constructor as new hypotheses.  (It also
    produced an auxiliary equality, which happens to be useless here.)
    We'll begin exploring this more general behavior of inversion in
    what follows. *)

(** **** Exercise: 1 star (inversion_practice) *)
Theorem SSSSev__even : forall n,
  ev (S (S (S (S n)))) -> ev n.
Proof.
  intros.
  inversion H. inversion H1.
  apply H3. Qed.

(** The [inversion] tactic can also be used to derive goals by showing
    the absurdity of a hypothesis. *)

Theorem even5_nonsense :
  ev 5 -> 2 + 2 = 9.
Proof.
  intros.
  inversion H. inversion H1. inversion H3.
  Qed.

(** **** Exercise: 3 stars, advanced (ev_ev__ev) *)
(** Finding the appropriate thing to do induction on is a
    bit tricky here: *)

Theorem ev_ev__ev : forall n m,
  ev (n+m) -> ev n -> ev m.
Proof.
  intros.
  induction H0. simpl in H. apply H.
  simpl in H. inversion H. apply IHev in H2.
  apply H2. Qed.

(** **** Exercise: 3 stars, optional (ev_plus_plus) *)
(** Here's an exercise that just requires applying existing lemmas.  No
    induction or even case analysis is needed, but some of the rewriting
    may be tedious. *)
(*Theorem ev_sum : forall n m,
   ev n -> ev m -> ev (n+m).

Theorem double_even : forall n,
  ev (double n).
*)
Theorem ev_plus_plus : forall n m p,
  ev (n+m) -> ev (n+p) -> ev (m+p).
Proof.
  intros. apply ev_ev__ev with (n := n+n) (m := m+p).
  replace (n+n + (m + p)) with ((n +m) + (n+p)).
  apply ev_sum. apply H. apply H0.
  rewrite -> plus_swap. rewrite -> plus_assoc. rewrite -> plus_assoc.
  rewrite -> plus_assoc. reflexivity.
  rewrite <- double_plus.
  apply double_even.
  Qed.

(* ####################################################### *)
(** * Additional Exercises *)

(** **** Exercise: 4 stars (palindromes) *)
(** A palindrome is a sequence that reads the same backwards as
    forwards.

    - Define an inductive proposition [pal] on [list X] that
      captures what it means to be a palindrome. (Hint: You'll need
      three cases.  Your definition should be based on the structure
      of the list; just having a single constructor
    c : forall l, l = rev l -> pal l
      may seem obvious, but will not work very well.)

    - Prove that
       forall l, pal (l ++ rev l).
    - Prove that
       forall l, pal l -> l = rev l.
*)
(*
Inductive ev : nat -> Prop :=
  | ev_0 : ev O
  | ev_SS : forall n:nat, ev n -> ev (S (S n)).
*)
(*
Inductive pal : list nat -> Prop :=
  |empty : pal nil
  |single : forall (x : nat), pal [x]
  |palCons : forall (x : nat) (l : list nat), pal l -> pal (x :: (l ++ [x])).
*)

Inductive pal : forall (X : Type) (l : list X), Prop :=
  |empty : forall X, pal X nil
  |single : forall X (x : X), pal X [x]
  |palCons : forall X (x : X) (l : list X), pal X l ->
                                            pal X (x :: (snoc l x))
.

(** **** Exercise: 5 stars, optional (palindrome_converse) *)
(** Using your definition of [pal] from the previous exercise, prove
    that
     forall l, l = rev l -> pal l.
*)

Theorem Pal_App_Rev : forall (X : Type) (l : list X), pal X (l ++ rev l).
Proof.
  intros.
  induction l.
  simpl. apply empty.
  simpl. Admitted.

Lemma Align : forall (X : Type) (x : X) (l l' : list X),
                x::l = rev l ++[x] -> x :: l' ++ [x] = x :: rev l' ++ [x].
Proof.
  induction l'.
  intros. simpl. reflexivity.
  intros. apply IHl' in H. Admitted.

Theorem revPal : forall (X : Type) (l : list X), l = rev l -> pal X l.
Proof.
  intros.
  induction l. apply empty.
  simpl in H. rewrite -> SnocApp in H. Admitted.

Inductive natBy2 : nat -> Prop :=
  |natBy2_0 : natBy2 0
  |natBy2_1 : natBy2 1
  |natBy2_SS : forall n, natBy2 n -> natBy2 (S(S n)).

Theorem palRev : forall (X : Type) (l : list X), pal X l -> l = rev l.
Proof.
  intros.

(* FILL IN HERE *)
(** [] *)

(** **** Exercise: 4 stars, advanced (subsequence) *)
(** A list is a _subsequence_ of another list if all of the elements
    in the first list occur in the same order in the second list,
    possibly with some extra elements in between. For example,
    [1,2,3]
    is a subsequence of each of the lists
    [1,2,3]
    [1,1,1,2,2,3]
    [1,2,7,3]
    [5,6,1,9,9,2,7,3,8]
    but it is _not_ a subsequence of any of the lists
    [1,2]
    [1,3]
    [5,6,2,1,7,3,8]

    - Define an inductive proposition [subseq] on [list nat] that
      captures what it means to be a subsequence. (Hint: You'll need
      three cases.)

    - Prove that subsequence is reflexive, that is, any list is a
      subsequence of itself.

    - Prove that for any lists [l1], [l2], and [l3], if [l1] is a
      subsequence of [l2], then [l1] is also a subsequence of [l2 ++
      l3].

    - (Optional, harder) Prove that subsequence is transitive -- that
      is, if [l1] is a subsequence of [l2] and [l2] is a subsequence
      of [l3], then [l1] is a subsequence of [l3].  Hint: choose your
      induction carefully!
*)

Inductive subseq : forall (l1 l2 : list nat), Prop :=
  |subseqNil : forall (l2 : list nat), subseq [] l2
  |subseqEq : forall (x : nat) (l1 l2 : list nat), subseq l1 l2 ->
                                                   subseq (x::l1) (x::l2)
  |subseqNE : forall (x : nat) (l1 l2 : list nat), subseq l1 l2 ->
                                                   subseq l1 (x::l2).

Theorem SubseqRefl : forall (l : list nat), subseq l l.
Proof.
  intros. induction l.
   apply subseqNil.
   apply subseqEq. apply IHl. Qed.

Theorem SubseqExtra : forall (l1 l2 l3 : list nat), subseq l1 l2 ->
                                                    subseq l1 (l2 ++ l3).
Proof.
  intros. induction H.
  apply subseqNil.
  simpl. apply subseqEq. apply IHsubseq.
  simpl. apply subseqNE. apply IHsubseq.
  Qed.

Theorem SubseqTrans : forall (l1 l2 l3 : list nat), subseq l1 l2 -> subseq l2 l3 ->
                                                    subseq l1 l3.
Proof.
  intros. generalize dependent l1.
  induction H0.
  { intros. destruct l1. apply subseqNil.
    inversion H. }
  { intros.

  induction H. intros. apply subseqNil.
  intros. Admitted.

(** **** Exercise: 2 stars, optional (R_provability) *)
(** Suppose we give Coq the following definition:
    Inductive R : nat -> list nat -> Prop :=
      | c1 : R 0 []
      | c2 : forall n l, R n l -> R (S n) (n :: l)
      | c3 : forall n l, R (S n) l -> R n l.
    Which of the following propositions are provable?

    - [R 2 [1,0]]      yes
    - [R 1 [1,2,1,0]]  yes
    - [R 6 [3,2,1,0]]  yes
*)

(** [] *)

(* $Date: 2013-07-01 18:48:47 -0400 (Mon, 01 Jul 2013) $ *)


Inductive bin : Type :=
  | Z
  | B0 (n : bin)
  | B1 (n : bin).

Fixpoint incr (m : bin) : bin :=
   match m with
   | Z => B1 Z
   | B0 n => B1 n
   | B1 n => B0 (incr n)
   end.

Fixpoint bin_to_nat (m:bin) : nat :=
  match m with
  | Z => O
  | B0 n => 2 * bin_to_nat n
  | B1 n => 1 + 2 * bin_to_nat n
  end.

Fixpoint nat_to_bin (n:nat) : bin :=
  match n with
  | O => Z
  | S n' => incr (nat_to_bin n')
  end.

Lemma n_plus_0_eq_n : forall n : nat,
  n + 0 = n.
Proof.
  intros n. induction n as [| n' IHn'].
  - reflexivity.
  - simpl. rewrite -> IHn'. reflexivity.
  Qed.

Lemma plus_n_Sm : forall n m : nat,
  S (n + m) = n + (S m).
Proof.
  intros n. induction n as [| n' IHn'].
  - simpl. reflexivity.
  - induction m as [| m' IHm'].
    + simpl. rewrite -> IHn'. reflexivity.
    + simpl. rewrite -> IHn'. reflexivity.
  Qed.

Lemma btn_incr_b_eq_S_btn_b : forall b : bin,
  bin_to_nat (incr b) = S(bin_to_nat b).
Proof.
  intros b. induction b as [| b' | b'].
  - reflexivity.
  - reflexivity.
  -
  simpl. rewrite -> n_plus_0_eq_n. rewrite -> n_plus_0_eq_n.
  rewrite -> IHb'.
  rewrite -> plus_n_Sm.
  reflexivity.
  Qed.

Theorem nat_bin_nat : forall n,
  bin_to_nat (nat_to_bin n) = n.
Proof.
    intros n. induction n as [| n' IHn'].
    - reflexivity.
    - simpl. rewrite -> btn_incr_b_eq_S_btn_b. rewrite -> IHn'. reflexivity.
    Qed.


(*************)
(* Induction *)
(*************)

Require Import Nat.

(*
  Let's prove that zero is a left identity
  for addition.
*)

Theorem zero_plus_n : forall n, 0 + n = n.
Proof.
  intro.
  simpl.
  reflexivity.
Qed.

(*
  Great, that was easy! Now let's prove
  that zero is also a right identity.
*)

Theorem n_plus_zero : forall n, n + 0 = n.
Proof.
  intro.
  simpl. (* Uh oh, nothing happened! *)
Abort.

(* Recall the definition of addition. *)

Print add.

(*
  From this, it's clear why 0 + n = n.
  But how do we prove n + 0 = n?
  We need induction.
*)

Check nat_ind.

(*
  Let's use that induction principle to
  prove that zero is a neutral element
  of addition.
*)

Theorem n_plus_zero : forall n, n + 0 = n.
Proof.
  intro.

  (*
    Instead of applying `nat_ind` directly,
    it is easier to use the `induction`
    tactic.
  *)
  induction n.

  (* Two subgoals are generated: *)
  - simpl.
    reflexivity.
  - simpl.
    rewrite IHn.
    reflexivity.
Qed.

Print n_plus_zero.

(* Let's prove addition is associative. *)

Theorem plus_assoc :
  forall n m p,
  n + (m + p) = (n + m) + p.
Proof.
  intros.
  induction n.
  - simpl.
    reflexivity.
  - simpl.
    rewrite IHn.
    reflexivity.
Qed.